*********************************
***ANOTHER DAY, ANOTHER DOLLAR***
********BESSEN & CONNELL*********
********REPLICATION FILE*********
*********************************


use "another_day_another_currency.dta", clear


**TABLE 1
global controls "i.ideology i.pid i.region wealth01 edu2 female trustleg natpride rural unemp retired remit pol_interest vote_correa otherparty"
//Model 1
probit opiniondollarbi eighteenin2000 $controls, cluster(region)
estimates store MODEL1, title((1))
//Model 2
probit opiniondollarbi c.eighteenin2000##c.wealth01 $controls, cluster(region)
estimates store MODEL2, title((2))
//Model 3
probit opiniondollarbi c.eighteenin2000##c.edu2 $controls, cluster(region)
estimates store MODEL3, title((3))
estout MODEL1 MODEL2 MODEL3, cells(b(star fmt(3)) se(par fmt(3))) ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) 


**FIGURE 1
*Figure 1A
qui probit opiniondollarbi c.eighteenin2000##c.wealth01 $controls, cluster(region)
qui margins, dydx(wealth01) at(eighteenin2000=(0(1)1)) atmean
qui marginsplot, xsize(10) ysize(9) yline(0) recast(scatter) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Wealth on Pro Dollar) xtitle("") title("") xscale(range(-0.5(0.5)1.5))
*Figure 1B
qui probit opiniondollarbi c.eighteenin2000##c.edu2 $controls, cluster(region)
qui margins, dydx(edu2) at(eighteenin2000=(0(1)1)) atmean
qui marginsplot, xsize(10) ysize(9) yline(0) recast(scatter) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Education on Pro Dollar) xtitle("") title("") xscale(range(-0.5(0.5)1.5))


**FIGURE 2
*Figure 2A
qui probit opiniondollarbi c.eighteenin2000##c.wealth01 $controls, cluster(region)
qui margins, at(wealth01=(0(.1)1) eighteenin2000=(0(1)1)) atmean
qui marginsplot, xsize(10) ysize(9) addplot(hist wealth01 if e(sample), yaxis(2) xscale(range(0(.1)1)) xlabel(0(.1)1) yscale(axis(2) range(0 10) off) lc(white%0) fcolor(sand%25) recast(rbar)) recast(line) recastci(rarea) ciopts(fc(blue%30) lc(white%0)) xtitle(Wealth Index) title("") ytitle(Predicted Support for Dollarization) graphregion(fcolor(white) lcolor(white))
*Figure 2B
qui probit opiniondollarbi c.eighteenin2000##c.edu2 $controls, cluster(region)
qui margins, at(edu2=(0(1)1) eighteenin2000=(0(1)1)) atmean
qui marginsplot, xsize(10) ysize(9) addplot(hist edu2 if e(sample), yaxis(2) xscale(range(-0.5(0.5)1.5)) xlabel(0(1)1) yscale(axis(2) range(0 100) off) lc(white%0) fcolor(sand%25) recast(rbar)) recast(connected) recastci(rcap) xtitle("") title("") ytitle(Predicted Support for Dollarization) graphregion(fcolor(white) lcolor(white))


**FIGURE 3
*Figure 3A
qui probit opiniondollarbi c.eighteenin2000##c.wealth01 $controls, cluster(region)
margins, dydx(eighteenin2000) at(wealth01=(0(.1)1)) atmean
marginsplot, yline(0) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Experience on Pro Dollar) title("") xtitle(Wealth) yscale(range(-.4(.1).1)) yscale(range(-.4(.1).1)) ylabel(-.4(.1).1)
*Figure 3B
qui probit opiniondollarbi c.eighteenin2000##c.edu2 $controls mother_edu, cluster(region)
margins, dydx(eighteenin2000) at(edu2=(0(1)1)) atmean
marginsplot, yline(0) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Experience on Pro Dollar) title("") xscale(range(-0.5(0.5)1.5)) recast(scatter) xtitle("") yscale(range(-.4(.1).1)) ylabel(-.4(.1).1)


******************************************************************************
*******************************APPENDIX***************************************
******************************************************************************

**TABLE A1 (Summary Statistics)
global controls "i.ideology i.pid i.region wealth01 edu2 female trustleg natpride rural unemp retired remit pol_interest vote_correa otherparty"
qui probit opiniondollarbi eighteenin2000 $controls, cluster(region)
sutex opiniondollarbi opiniondollar eighteenin2000 wealth01 edu2 age female trustleg natpride rural unemp retired remit pol_interest vote_correa otherparty left right center mother_edu if e(sample), minmax nobs digits(3)
//Placebo DVs:
qui reg pref_stateown c.eighteenin2000 $controls, cluster(region)
sutex pref_stateown if e(sample), minmax nobs digits(3)
qui reg reduce_ineq c.eighteenin2000 $controls, cluster(region)
sutex reduce_ineq if e(sample), minmax nobs digits(3)


**TABLE A2 (Order Probit Models)
oprobit opiniondollar eighteenin2000 $controls, cluster(region)
estimates store MODELA1, title((A1))
oprobit opiniondollar c.eighteenin2000##c.wealth01 $controls, cluster(region)
estimates store MODELA2, title((A2))
oprobit opiniondollar c.eighteenin2000##c.edu2 $controls, cluster(region)
estimates store MODELA3, title((A3))
estout MODELA1 MODELA2 MODELA3, cells(b(star fmt(3)) se(par fmt(3))) ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) ///
	stats(N, fmt(0 0 3) label(Observations Countries R$^2$)) label


**TABLE A3 (Sensitivity Checks: Wealth)
probit opiniondollarbi c.eighteenin2003##c.wealth01 $controls, cluster(region)
estimates store MODELA4, title((A4))
probit opiniondollarbi c.eighteenin2002##c.wealth01 $controls, cluster(region)
estimates store MODELA5, title((A5))
probit opiniondollarbi c.eighteenin2001##c.wealth01 $controls, cluster(region)
estimates store MODELA6, title((A6))
probit opiniondollarbi c.eighteenin2000##c.wealth01 $controls, cluster(region)
estimates store MODELA7, title((A7))
estout MODELA4 MODELA5 MODELA6 MODELA7, cells(b(star fmt(3)) se(par fmt(3))) ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**TABLE A4 (Sensitivity Checks: Education)
*Education
probit opiniondollarbi c.eighteenin2003##c.edu2 $controls, cluster(region)
estimates store MODELA8, title((A8))
probit opiniondollarbi c.eighteenin2002##c.edu2 $controls, cluster(region)
estimates store MODELA9, title((A9))
probit opiniondollarbi c.eighteenin2001##c.edu2 $controls, cluster(region)
estimates store MODELA10, title((A10))
probit opiniondollarbi c.eighteenin2000##c.edu2 $controls, cluster(region)
estimates store MODELA11, title((A11))
estout MODELA8 MODELA9 MODELA10 MODELA11, cells(b(star fmt(3)) se(par fmt(3)))  ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**FIGURE 4
*Figure 4A
coefplot MODELA4 MODELA5 MODELA6 MODELA7, keep(c.eighteenin2003#c.wealth01 c.eighteenin2002#c.wealth01 c.eighteenin2001#c.wealth01 c.eighteenin2000#c.wealth01) xline(0) xscale(range(-0.25(0.5)2)) graphregion(fcolor(white) lcolor(white)) xtitle(Experience X Wealth Coefficient)
*Figure 4B
coefplot MODELA8 MODELA9 MODELA10 MODELA11, keep(c.eighteenin2003#c.edu2 c.eighteenin2002#c.edu2 c.eighteenin2001#c.edu2 c.eighteenin2000#c.edu2) xline(0) xscale(range(-0.1(0.1).5)) graphregion(fcolor(white) lcolor(white)) xtitle(Experience X Education Coefficient)


**TABLE A5 (Control for Age)
probit opiniondollarbi eighteenin2000 age $controls, cluster(region)
estimates store MODELA12, title((A12))
probit opiniondollarbi c.eighteenin2000##c.wealth01 age $controls, cluster(region)
estimates store MODELA13, title((A13))
probit opiniondollarbi c.eighteenin2000##c.edu2 age $controls, cluster(region)
estimates store MODELA14, title((A14))
estout MODELA12 MODELA13 MODELA14, cells(b(star fmt(3)) se(par fmt(3))) ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**TABLE A6 (Continous Education Variable)  - DOESNT MATCH WITH APPENDIX 
global controls2 "wealth01 edu female trustleg natpride rural unemp retired remit pol_interest vote_correa otherparty i.ideology i.pid i.region"
probit opiniondollarbi eighteenin2000 $controls2, cluster(region)
estimates store MODELA15, title((A15))
probit opiniondollarbi c.eighteenin2000##c.edu $controls2, cluster(region)
estimates store MODELA16, title((A16))
oprobit opiniondollar eighteenin2000 $controls2, cluster(region)
estimates store MODELA17, title((A17))
oprobit opiniondollar c.eighteenin2000##c.edu $controls2, cluster(region)
estimates store MODELA18, title((A18))
estout MODELA15 MODELA16 MODELA17 MODELA18, cells(b(star fmt(3)) se(par fmt(3)))   ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**TABLE A7 (Placebo Checks)
reg pref_stateown c.eighteenin2000##c.wealth01 $controls, cluster(region)
estimates store MODELA19, title((A19))
reg pref_stateown c.eighteenin2000##c.edu2 $controls, cluster(region)
estimates store MODELA20, title((A20))
reg reduce_ineq c.eighteenin2000##c.wealth01 $controls, cluster(region)
estimates store MODELA21, title((A21))
reg reduce_ineq c.eighteenin2000##c.edu2 $controls, cluster(region)
estimates store MODELA22, title((A22))
estout MODELA19 MODELA20 MODELA21 MODELA22, cells(b(star fmt(3)) se(par fmt(3)))  ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label

**TABLE A8 (Main Models but Controls Reported)
probit opiniondollarbi eighteenin2000 $controls, cluster(region)
estimates store MODELA23, title((A23))
probit opiniondollarbi c.eighteenin2000##c.wealth01 $controls, cluster(region)
estimates store MODELA24, title((A24))
probit opiniondollarbi c.eighteenin2000##c.edu2 $controls, cluster(region)
estimates store MODELA25, title((A25))
estout MODELA23 MODELA24 MODELA25, cells(b(star fmt(3)) se(par fmt(3)))  ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3)  ///
	label(Observations Countries R$^2$)) label


**TABLE A9 (Effects of Mother's Education)
probit opiniondollarbi eighteenin2000 mother_edu $controls, cluster(region)
estimates store MODELA26, title((A26))
probit opiniondollarbi c.eighteenin2000##c.wealth01 mother_edu $controls, cluster(region)
estimates store MODELA27, title((A27))
probit opiniondollarbi c.eighteenin2000##c.edu2 mother_edu $controls, cluster(region)
estimates store MODELA28, title((A28))
estout MODELA26 MODELA27 MODELA28, cells(b(star fmt(3)) se(par fmt(3))) ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**TABLE A10 (Interactive Effects with Vote Choice)
probit opiniondollarbi c.vote_correa##c.wealth eighteenin2000 $controls, cluster(region)
estimates store MODELA29, title((A29))
probit opiniondollarbi c.vote_correa##c.edu2 eighteenin2000 $controls, cluster(region)
estimates store MODELA30, title((A30))
estout MODELA29 MODELA30, cells(b(star fmt(3)) se(par fmt(3))) ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**TABLE A11 (Conditional Effects of Unemployment)
probit opiniondollarbi c.eighteenin2000##c.unemp $controls, cluster(region)
estimates store MODELA31, title((A31))
estout MODELA31, cells(b(star fmt(3)) se(par fmt(3))) ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**TABLE A12
probit opiniondollarbi c.eighteenin2000##c.wealth01##c.pol_interest $controls, cluster(region)
estimates store MODELA32, title((A32))
probit opiniondollarbi c.eighteenin2000##c.edu2##c.pol_interest $controls, cluster(region)
estimates store MODELA33, title((A33))
estout MODELA32 MODELA33, cells(b(star fmt(3)) se(par fmt(3)))  ///
	starlevels($^{+}$ 0.10 $^{*}$ 0.05 $^{**}$ 0.01 $^{***}$ 0.001) stats(N N_g, fmt(0 0 3) ///
	label(Observations Countries R$^2$)) label


**FIGURE A1 (Past Vote Choice)
qui probit opiniondollarbi eighteenin2000 $controls, cluster(region)
margins, at(vote_correa==1 otherparty==0) at(vote_correa==0 otherparty==1) atmean
qui marginsplot, graphregion(fcolor(white) lcolor(white)) ytitle(Predicted Support for Dollarization) title("") xscale(range(0(1)3)) recast(scatter) xtitle("")


**FIGURE A2
qui probit opiniondollarbi c.eighteenin2000##c.unemp $controls, cluster(region)
qui margins, dydx(unemp) at(eighteenin2000=(0(1)1)) atmean
qui marginsplot, yline(0) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Unemployment on Pro Dollar) title("") xscale(range(-0.5(0.5)1.5)) recast(scatter) xtitle("")


**FIGURE A3 (Triple Interaction with Wealth & Political Interest)
qui probit opiniondollarbi c.eighteenin2000##c.wealth01##c.pol_interest $controls, cluster(region)
qui margins, dydx(wealth01) at(eighteenin2000=(0(1)1) pol_interest==1) atmean
qui marginsplot, xsize(10) ysize(9) yline(0) recast(scatter) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Wealth on Pro Dollar) xtitle("") title("") xscale(range(-0.5(0.5)1.5))
qui margins, dydx(wealth01) at(eighteenin2000=(0(1)1) pol_interest==4) atmean
qui marginsplot, xsize(10) ysize(9) yline(0) recast(scatter) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Wealth on Pro Dollar) xtitle("") title("") xscale(range(-0.5(0.5)1.5))


**FIGURE A4 (Triple Interaction with Edu & Political Interest)
qui probit opiniondollarbi c.eighteenin2000##c.edu2##c.pol_interest $controls, cluster(region)
qui margins, dydx(edu2) at(eighteenin2000=(0(1)1) pol_interest==1) atmean
qui marginsplot, xsize(10) ysize(9) yline(0) recast(scatter) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Education on Pro Dollar) xtitle("") title("") xscale(range(-0.5(0.5)1.5))
qui margins, dydx(edu2) at(eighteenin2000=(0(1)1) pol_interest==4) atmean
qui marginsplot, xsize(10) ysize(9) yline(0) recast(scatter) graphregion(fcolor(white) lcolor(white)) ytitle(Marginal Effect of Education on Pro Dollar) xtitle("") title("") xscale(range(-0.5(0.5)1.5))
